ASSIGNMENT 4 – PROJECT 2
FESTO PROJECT

282.778 – Mechatronics

Caelan Vandermey

Table of Contents

Table of Figures

1. Introduction

This project is being conducted on a FESTO Modular Production System that uses systems one, two, four and nine to illustrate the uses ability to use a PC and PLC to control mechatronic systems using pneumatic, mechanical and electrical actuators and signal conditioning. This setup is shown in Figure 1.



Figure 1 Massey University FESTO Production System

During the time that the project is being run, station 3 (system four) was inoperable and resulted in the setup not being able to properly run with interstation communication. This has resulted in stations being programmed as if they are not connected (unlike what is shown in Figure 1). This alters the objectives from what is shown below:

- Program a PLC to transport plastic objects from one side of the FESTO Distribution Station to the other.
- Program a PLC to transport plastic objects from one side of the FESTO Testing Station to the other.
- Program a PLC to transport plastic objects from one side of the FESTO Handling Station to the other.
- Program a PLC to sort objects into different groups using the FESTO Sorting Station.

To the following:

- Program a PLC to transport plastic objects from one side of the FESTO Distribution Station to the other.
- Program a PLC to transport plastic objects from one side of the FESTO Testing Station to the other.
- Create a program that should be capable of transporting plastic objects from one side of the FESTO Handling Station to the other.
- Program a PLC to sort objects into different groups using the FESTO Sorting Station.

This is what will be conducted and explained in this report.

2. Methodology

2.1. Standard Main Code

Figure 2 shows the standard component to the main code. This is the same for all four stations. Each station has a section of reset code that located after this that handle the resetting of the station.

The first if statement is checking for the emergency stop button to be pressed. This has push to break logic as a safety feature, meaning that there is a current until the button is pressed. This will activate the emergency program that was already built into the project when given to the class.

The second if statement looks for no operation being taken and begins the emergency blink code to indicate that nothing is happening.

The third and fifth if statements check for the stop button being pressed and will run the stop code that was given to the class. This button also has push to break logic so that if power is lost to the button, the controller will turn the equipment off. The fifth if statement also specifically ends the sequence program.

The fourth if statement checks for the start button to be pressed and then begins the sequence program if this is true.

```
IF            N      Em_Stop          'Emergency stop unlocked
      OR             FI
THEN  CMP 0                           'Emergency program

IF                   NOP
THEN  CMP 2                           'Emergency Blink

IF            N      S2               'Stop button (normally closed)
THEN  CMP 1                           'Stop program

IF                   S1               'Start button
THEN  SET            P1

IF            N      S2               'Stop button (normally closed)
THEN  CMP 1                           'Stop program
      RESET          P1
```

Figure 2 Standard Main Code

2.2. Station One (01_Distributing)

2.2.1. Main

Figure 3 shows the reset code in the main program for station one. This checks if the reset button is pressed and then drives the swivel arms to the subsequent station while turning off the vacuum for the workpiece movement and retracting the ejecting cylinder.

```
IF                   S4               'Reset button
THEN  RESET          _3M1             'Swivel drive to magazine
      SET            _3M2             'Swivel drive to subsequent station
      RESET          _2M1             'Vacuum on
      RESET          _1M1             'Ejecting cylinder push out workpiece
```

Figure 3 Station One Main Reset Code

2.2.2. Sequence

2.2.2.1. Step One

Figure 4 shows the first step in station one's sequence program. This has

The first if statement ensures that the arm is moved to the subsequent station. This is done by checking if the arm is not in the subsequent station, and if this is true, setting and resetting the appropriate drive outputs. This is done so that when the workpiece is ejected from the magazine, the arm will not be blocking the path of the workpiece.

The second if statement waits until the arm is in the subsequent station and for at least one workpiece to be loaded into the magazine. This then moves on to the second step. If the check returns as false, then the ejection impulse is turned off.

```
STEP 1
 IF            N       _3B2          'Swivel drive i. pos. subsequ. stat.
 THEN  RESET           _3M1          'Swivel drive to magazine
       SET             _3M2          'Swivel drive to subsequent station

 IF                    _3B2          'Swivel drive i. pos. subsequ. stat.
       AND     N       _B4           'Magazine empty
 THEN  JMP TO 2
 OTHRW RESET           _2M2          'Ejection impulse on
```

Figure 4 Station One Sequence Step One Code

2.2.2.2. Step Two

Figure 5 shows the second step of station one's sequence program. This ensures that the arm is in the subsequent station and that the magazine is not empty before ejecting a workpiece from the magazine.

If the magazine is empty, then the program will jump back to step one to wait for a new workpiece.

```
STEP 2
 IF                    _3B2          'Swivel drive i. pos. subsequ. stat.
       AND     N       _B4           'Magazine empty
 THEN  SET             _1M1          'Ejecting cylinder push out workpiece
       JMP TO 3

 IF                    _B4           'Magazine empty
 THEN  JMP TO 1
```

Figure 5 Station One Sequence Step Two Code

2.2.2.3. Step Three

Figure 6 shows the code for step three of station one's sequence code. This checks that a workpiece has been pushed out and then moves the arm to the magazine station while retracting the workpiece ejector and then moves on to the next step.

```
STEP 3
 IF                    _1B1          'Ejecting cylinder retractet (p o)
 THEN  RESET           _3M2          'Swivel drive to subsequent station
       SET             _3M1          'Swivel drive to magazine
       RESET           _1M1          'Ejecting cylinder push out workpiece
       JMP TO 4
```

Figure 6 Station One Sequence Step Three Code

2.2.2.4. Step Four

Figure 7 shows the step four code of station one where the if statement waits for the workpiece pusher to retract (so that the workpiece can be moved) and the arm to move to the magazine station. Once those conditions are met, the system turns the vacuum on (to hold the workpiece), moves the arm to the subsequent station and moves on to the final step in the sequent code.

```
STEP 4
 IF                    _1B2          'Ejecting cylinder extended (p i)
       AND             _3B1          'Swivel drive in pos. magazine
 THEN  SET             _2M1          'Vacuum on
       RESET           _3M1          'Swivel drive to magazine
       SET             _3M2          'Swivel drive to subsequent station
       JMP TO 5
```

Figure 7 Station One Sequence Step Four Code

2.2.2.5. Step Five

Figure 8 shows station one's code for step five, where the system waits until the arm is in the subsequent station. Then turns off the vacuum, turns on the ejection impulse (which removes the possibility of a vacuum still holding the workpiece on the arm) and jumps to step two, where the code can then loop through.

```
STEP 5
 IF                    _3B2          'Swivel drive i. pos. subsequ. stat.
 THEN  RESET           _2M1          'Vacuum on
       SET             _2M2          'Ejection impulse on
       JMP TO 2
```

Figure 8 Station One Sequence Step Five Code

2.3. Station Two (02_Testing)

2.3.1. Main

Figure 9 shows the reset code for the main code in station two. This lowers the lifting cylinder and retracts the ejecting arm.

```
IF                    S4            'Reset button
THEN  RESET           _2M1          'Extend ejecting cylinder
      RESET           _1M2          'Raise lifting cylinder
      SET             _1M1          'Lower lifting cylinder
```

Figure 9 Station Two Main Reset Code

2.3.2. Sequence

2.3.2.1. Step One

Figure 10 shows the code for step on in station two. The first if statement checks if the lifting system is not lowered and lowers it if it is not. The second waits for a part to be placed in the tray and the lifting cylinder to be lowered, where it will then start a timer for two seconds and moves on to step two.

```
STEP 1
 IF            N       _1B2          'Lifting cylinder lowered
 THEN  RESET           _1M2          'Raise lifting cylinder
       SET             _1M1          'Lower lifting cylinder

 IF                    Part_AV       'Workpiece available
       AND             _1B2          'Lifting cylinder lowered
 THEN  SET             T1            'Blink timer 1
         WITH          2s
       JMP TO 2
```

Figure 10 Station Two Sequence Step One Code

2.3.2.2. Step Two

Figure 11 shows the step two code for station two. This waits for the set timer to expire (two seconds) and then begins to raise the workpiece and move on to step three.

```
STEP 2
 IF            N       T1              'Blink timer 1
 THEN  RESET           _1M1            'Lower lifting cylinder
       SET             _1M2            'Raise lifting cylinder
       JMP TO 3
```

Figure 11 Station Two Sequence Step Two Code

2.3.2.3. Step Three

Figure 12 shows the step three code for station two. This waits for the lifting cylinder to be raised and then ejects the work piece onto the air slide. The air slide is also turned on (making it like an air hockey table), a timer is set for two seconds and the code jumps to step four.

```
STEP 3
 IF                    _1B1            'Lifting cylinder raised
 THEN  SET             _2M1            'Extend ejecting cylinder
       SET             _3M1            'air slide on
       SET             T2              'Blink timer 2
         WITH          2s
       JMP TO 4
```

Figure 12 Station Two Sequence Step Three Code

2.3.2.4. Step Four

Figure 13 shows the step four code for station two. This waits for the timer to expire (two seconds), retracts the ejecting pusher and moves onto the final step in the sequence code.

```
STEP 4
 IF            N       T2              'Blink timer 2
 THEN  RESET           _2M1            'Extend ejecting cylinder
       JMP TO 5
```

Figure 13 Station Two Sequence Step Four Code

2.3.2.5. Step Five

Figure 14 shows station two's step five code. Where the system waits for the ejecting pusher to retract and then begins to lower the lifting cylinder. The code then loops back to step one.

```
STEP 5
 IF                    _2B1            'Ejecting cylinder retracted
 THEN  RESET           _1M2            'Raise lifting cylinder
       SET             _1M1            'Lower lifting cylinder
       JMP TO 1
```

Figure 14 Station Two Sequence Step Five Code

2.4. Station Three (04_Handling)

2.4.1. Main

Figure 15 shows the reset code for station three's main program. This moves the system to the upstream station and allows the gripper to retract.

```
IF            N      S4              'Reset button
THEN  RESET          _1M2            'Handling to downstream station
      SET            _1M1            'Handling to upstream station
      RESET          _2M1            'Extend gripper
```

Figure 15 Station Three Main Reset Code

2.4.2. Sequence

2.4.2.1. Step One

Figure 16 shows station three's step one code where the handling system is moved to the upstream location if it is not already and allows the gripper to retract (again, if not already). Once both criteria are met, the system will move on to step two.

```
STEP 1
 IF             N      _1B1            'Handling at upstream station
 THEN  RESET           _1M2            'Handling to downstream station
       SET             _1M1            'Handling to upstream station

 IF             N      _2B2            'Gripper retracted
 THEN  RESET           _2M1            'Extend gripper

 IF                    _1B1            'Handling at upstream station
       AND             _2B2            'Gripper retracted
 THEN  JMP TO 2
```

Figure 16 Station Three Sequence Step One Code

2.4.2.2. Step Two

Figure 17 shows the step two code for station three. This waits for a part to be available to grab and then opens the gripper, extends it downwards and moves to step three.

```
STEP 2
 IF                    Part_AV         'Workpiece available
 THEN  SET             _3M1            'Open gripper
       SET             _2M1            'Extend gripper
       JMP TO 3
```

Figure 17 Station Three Sequence Step Two Code

2.4.2.3. Step Three

Figure 18 shows station three's step three where once the gripper is extended, the gripper is closed and retracted upwards. A timer is set with half a second, so that the workpiece is lifted far enough to move sideways, and the system moves on to step three.

```
STEP 3
 IF                    _2B1            'Gripper extended
 THEN  RESET           _3M1            'Open gripper
       RESET           _2M1            'Extend gripper
       SET             T1              'Blink timer 1
         WITH          0.5s
       JMP TO 4
```

Figure 18 Station Three Sequence Step Three Code

2.4.2.4. Step Four

Figure 19 shows step four of station three. The system waits for the gripper to be retracted, the timer to expire and checks that the handling system is not downstream. If these conditions are met, the handling system moves to the downstream location.

```
STEP 4
 IF                    _2B2          'Gripper retracted
       AND     N       _1B2          'Handling at downstream station
       AND     N       T1            'Blink timer 1
 THEN  RESET           _1M1          'Handling to upstream station
       SET             _1M2          'Handling to downstream station
       JMP TO 5
```

Figure 19 Station Three Sequence Step Four Code

2.4.2.5. Step Five

Figure 20 shows station three's step five. This waits for the handling system to be at the downstream station, then extends the gripper and moves on to step six.

```
STEP 5
 IF                    _1B2          'Handling at downstream station
 THEN  SET             _2M1          'Extend gripper
       JMP TO 6
```

Figure 20 Station Three Sequence Step Five Code

2.4.2.6. Step Six

Figure 21 shows step six for station three. This waits for the gripper to be extended and the handling system at the downstream position. Once those criteria are met the gripper is opened and a timer is set with half a second (so that the part can fall without unexpected interference). The system then moves on to step seven.

```
STEP 6
 IF                    _1B2          'Handling at downstream station
       AND             _2B1          'Gripper extended
 THEN  SET             _3M1          'Open gripper
       SET             T1            'Blink timer 1
         WITH          0.5s
       JMP TO 7
```

Figure 21 Station Three Sequence Step Six Code

2.4.2.7. Step Seven

Figure 22 shows the code for station three's step seven. This waits for the timer to expire and then loops to step one where the handling system will be moved to the upstream positon and the gripper will be retracted.

```
STEP 7
 IF            N       T1            'Blink timer 1
 THEN  JMP TO 1
```

Figure 22 Station Three Sequence Step Seven Code

2.5. Station Four (09_Sorting)

2.5.1. Main

Figure 23 shows the reset code for station four where all the outputs that are used are reset to their initial positions.

```
IF                  S4              'Reset button
THEN  RESET         _1M1            'Extend switch 1
      RESET         _2M1            'Extend switch 2
      RESET         _3M1            'Retract stopper
      RESET         _K1             'belt motor on
```

Figure 23 Station Four Main Reset Code

2.5.2. Sequence

2.5.2.1. Step One

Figure 24 shows station four's step one. This waits for a workpiece to be available and checks that the slides are not full. If these two criteria are met, then the belt is turned on and a timer is set for two seconds. The system then moves on to step two.

```
STEP 1
 IF                 Part_AV         'Part available
       AND     N    B4              'Slide full
 THEN  SET          _K1             'belt motor on
       SET          T1              'Blink timer 1
         WITH       2s
       JMP TO 2
```

Figure 24 Station Four Sequence Step One Code

2.5.2.2. Step Two

Figure 25 shows the step two code for station four. This step waits for the timer to expire (two seconds) and then moves on to step three.

```
STEP 2
 IF           N     T1              'Blink timer 1
 THEN  JMP TO 3
```

Figure 25 Station Four Sequence Step Two Code

2.5.2.3. Step Three

Figure 26 shows step three for station four. This has three different logic sets, one for each type of workpiece (red, metallic and black).

The first if statement handles red workpieces, where the colour will show up as not black and it will be non-metallic. This moves out the switch one arm, which will allow the red piece to fall onto the red slide (first slide). The stopper is then retracted and the system moves to step four.

The second if statement handles metallic workpieces, where the colour will show up as black and it will be metallic. This moves out the switch two arm, which will allow the metallic piece to fall onto the metallic slide (middle slide). The stopper is then retracted and the system moves to step four.

The third if statement handles black workpieces, where the colour will show up as black and it will be non-metallic. This does not move out either switch arm, which will allow the black piece to fall onto the black slide (far slide). The stopper is then retracted and the system moves to step four.

```
STEP 3
 IF                    B3              'Workpiece not black
       AND      N      B2              'Metallic workpiece
 THEN  SET            _1M1             'Extend switch 1
       SET            _3M1             'Retract stopper
       JMP TO 4

 IF                    B2              'Metallic workpiece
       AND      N      B3              'Workpiece not black
 THEN  SET            _2M1             'Extend switch 2
       SET            _3M1             'Retract stopper
       JMP TO 4

 IF             N      B3              'Workpiece not black
       AND      N      B2              'Metallic workpiece
 THEN  SET            _3M1             'Retract stopper
       JMP TO 4
```

Figure 26 Station Four Sequence Step Three Code

2.5.2.4. Step Four

Figure 27 shows the final step in station four's sequence code. This waits for the slide full sensor to be triggered (as this will happen when the workpiece falls onto a slide) and will reset the belt, switch arms and allow the stopper to extend again. The code then moves to step one so that it may loop.

```
STEP 4
 IF                    B4              'Slide full
 THEN  RESET          _K1              'belt motor on
       RESET          _1M1             'Extend switch 1
       RESET          _2M1             'Extend switch 2
       RESET          _3M1             'Retract stopper
       JMP TO 1
```

Figure 27 Station Four Sequence Step Four Code

3. Conclusion

In conclusion, the setup that Massey University uses is not what FESTO has created. FESTO has created many more stations than the four that were used and the ones that were used are not consecutive (kits one, two, four and nine are used) and station three is non-functional at this time. This results in code that deviates from what the handbooks explain to be the best solution.